

Chapter 48

Tests for Special Causes

Chapter Table of Contents

STANDARD TESTS FOR SPECIAL CAUSES	1731
Requesting Standard Tests	1734
Interpreting Standard Tests for Special Causes	1736
Modifying Standard Tests for Special Causes	1737
Applying Tests with Varying Subgroup Sample Sizes	1738
Labeling Signaled Points with a Variable	1740
Applying Tests with Multiple Phases	1741
Applying Tests with Multiple Sets of Control Limits	1742
Enhancing the Display of Signaled Tests	1745
NONSTANDARD TESTS FOR SPECIAL CAUSES	1746
Applying Tests to Range and Standard Deviation Charts	1746
Applying Tests Based on Generalized Patterns	1747
Customizing Tests with DATA Step Programs	1750

Chapter 48

Tests for Special Causes

This chapter provides details concerning standard and nonstandard tests for special causes that you can apply with the SHEWHART procedure.

Standard Tests for Special Causes

The SHEWHART procedure provides eight standard *tests for special causes*, also referred to as *rules for lack of control*, *supplementary rules*, *runs tests*, *runs rules*, *pattern tests*, and *Western Electric rules*. These tests improve the sensitivity of the Shewhart chart to small changes in the process.* You can also improve the sensitivity of the chart by increasing the rate of sampling, increasing the subgroup sample size, and using control limits that represent less than three standard errors of variation from the central line. However, increasing the sampling rate and sample size is often impractical, and tightening the control limits increases the chances of falsely signaling an out-of-control condition. By detecting particular nonrandom patterns in the points plotted on the chart, the tests can provide greater sensitivity and useful diagnostic information while incurring a reasonable probability of a false signal.

The patterns detected by the eight standard tests are defined in Table 48.1 and Table 48.2, and they are illustrated in Figure 48.1 and Figure 48.2. All eight tests were developed for use with fixed 3σ limits. The tests are indexed according to the numbering sequence used by Nelson (1984, 1985). You can request any combination of the eight tests by specifying the test *indexes* with the TESTS= option in the BOXCHART, CCHART, IRCHART, MCHART, MRCHART, NPCHART, PCHART, UCHART, XCHART, XRCHART, and XSCHART statements.

The following restrictions apply to the tests:

- Only Tests 1, 2, 3, and 4 are recommended for c charts, np charts, p charts, and u charts created with the CCHART, NPCHART, PCHART, and UCHART statements, respectively. In these four cases, Test 2 should not be used unless the process distribution is symmetric or nearly symmetric.
- By default, the TESTS= option is not applied with control limits that are not 3σ limits or that vary with subgroup sample size. You can use the NO3SIGMACHECK option to request tests for special causes when the SIGMAS= option specifies control limits other than 3σ limits. This is not recommended for standard control chart applications, since the standard tests for special causes are based on 3σ limits. You can apply test for special causes when control limits vary with subgroup sample size by using the LIMITN= or TESTNMETHOD= options (see page 1735 and page 1738).

*Cumulative sum control charts and moving average control charts also detect small shifts more quickly than an ordinary Shewhart chart. See Part 2, “The CUSUM Procedure,” and Part 5, “The MACONTROL Procedure.”

Table 48.1. Definitions of Tests 1 to 4

Test Index	Pattern Description
1	One point beyond Zone A (outside the control limits)
2	Nine points in a row in Zone C or beyond on one side of the central line (see Note 1 below)
3	Six points in a row steadily increasing or steadily decreasing (see Note 2 below)
4	Fourteen points in a row alternating up and down




Figure 48.1. Examples of Tests 1 to 4

Notes:

1. The number of points in Test 2 can be specified as 7, 8, 9, 11, 14, or 20 with the TEST2RUN= option.
2. The number of points in Test 3 can be specified as 6, 7, or 8 with the TEST3RUN= option.

Table 48.2. Definitions of Tests 5 to 8

Test Index	Pattern Description
5	Two out of three points in a row in Zone A or beyond
6	Four out of five points in a row in Zone B or beyond
7	Fifteen points in a row in Zone C on either or both sides of the central line
8	Eight points in a row on either or both sides of the central line with no points in Zone C




Figure 48.2. Examples of Tests 5 to 8

Requesting Standard Tests

See SHWTSC1
in the SAS/QC
Sample Library

The following example illustrates how to request the standard tests for special causes. The tests are applied to an \bar{X} chart for assembly offset measurements whose subgroup means, ranges, and sample sizes are provided by the variables OFFSETX, OFFSETR, and OFFSETN, respectively, in a data set named ASSEMBLY.*

```

data assembly;
  length system $ 1 comment $ 16;
  label sample = 'Sample Number';
  input system sample offsetx offsetr offsetn comment $16. ;
datalines;
T   1  19.80  3.8  5
T   2  17.16  8.3  5
T   3  20.11  6.7  5
T   4  20.89  5.5  5
T   5  20.83  2.3  5
T   6  18.87  2.6  5
T   7  20.84  2.3  5
T   8  23.33  5.7  5  New Tool
T   9  19.21  3.5  5
T  10  20.48  3.2  5
T  11  22.05  4.7  5
T  12  20.02  6.7  5
T  13  17.58  2.0  5
T  14  19.11  5.7  5
T  15  20.03  4.1  5
R  16  20.56  3.7  5  Changed System
R  17  20.86  3.3  5
R  18  21.10  5.6  5  Reset Tool
R  19  19.05  2.7  5
R  20  21.76  2.8  5
R  21  21.76  6.4  5
R  22  20.54  4.8  5
R  23  20.04  8.2  5
R  24  19.94  8.8  5
R  25  20.70  5.1  5
Q  26  21.40 12.1  7  Bad Reading
Q  27  21.32  3.2  7
Q  28  20.03  5.2  7  New Gauge
Q  29  22.02  5.9  7
Q  30  21.32  4.3  7
;

```

The following statements use the TESTS= option to request Tests 1 to 4. Note that the *process* OFFSET is specified in the XRCHART statement to indicate that the three summary variables OFFSETX, OFFSETR, and OFFSETN are to be read from the HISTORY= data set ASSEMBLY.

*The data set ASSEMBLY is also used by subsequent examples in this chapter.

```

title 'Analysis of Assembly Data';
symbol v=dot;

proc shewhart history=assembly;
  xrchart offset * sample / mu0      = 20
                                sigma0 = 2.24
                                limitn  = 5
                                alln
                                tests    = 1 to 4
                                ltests   = 2
                                vaxis    = 16 to 26 by 2
                                split    = '/' ;
  label offsetx = 'Avg Offset (cm)/Range';
run;

```

The chart is displayed in Figure 48.3. Test 1 is positive at the 8th subgroup, and Test 3 is positive at the 18th subgroup.




Figure 48.3. Standard Tests Using the TESTS= Option

The control limits in Figure 48.3 are based on standard values for the process mean and standard deviation specified with the MU0= and SIGMA0= options, respectively. Although the subgroup sizes vary, fixed control limits are displayed corresponding to a nominal sample size of five, which is specified with the LIMITN= option. Since ALLN is specified, points are displayed for all subgroups, regardless of sample size.

Note: If the LIMITN= option were not specified, the control limits would vary with subgroup sample size, and by default the tests would not be applied. An alternative method for applying the tests with varying subgroup sample sizes is discussed in “Applying Tests with Varying Subgroup Sample Sizes” on page 1738.

Interpreting Standard Tests for Special Causes

Nelson (1984, 1985) makes the following comments concerning the interpretation of the tests:

- When a process is in statistical control, the chance of a false signal for each test is less than five in one thousand.
- Test 1 is positive if there is a shift in the process mean, if there is an increase in the process standard deviation, or if there is a “single aberration in the process such as a mistake in calculation, an error in measurement, bad raw material, a breakdown of equipment, and so on” (Nelson 1985).
- Test 2 signals a shift in the process mean. The use of nine points (rather than seven as in Grant and Leavenworth 1988) for the pattern that defines Test 2 makes the chance of a false signal comparable to that of Test 1. (To control the number of points for the pattern in test 2, use the TEST2RUN= option in the chart statement.)
- Test 3 signals a drift in the process mean. Nelson (1985) states that causes can include “tool wear, depletion of chemical baths, deteriorating maintenance, improvement in skill, and so on.”
- Test 4 signals “a systematic effect such as produced by two machines, spindles, operators or vendors used alternately” (Nelson 1985).
- Tests 1, 2, 3, and 4 should be applied routinely; the combined chance of a false signal from one or more of these tests is less than one in a hundred. Nelson (1985) describes these tests as “a good set that will react to many commonly occurring special causes.”
- In the case of charts for variables, the first four tests should be augmented by Tests 5 and 6 when earlier warning is desired. The chance of a false signal increases to two in a hundred.
- Tests 7 and 8 indicate stratification (observations in a subgroup have multiple sources with different means). Test 7 is positive when the observations in the subgroup always have multiple sources. Test 8 is positive when the subgroups are taken from one source at a time.

Nelson (1985) also comments that “the probabilities quoted for getting false signals should not be considered to be very accurate” since the probabilities are based on assumptions of normality and independence that may not be satisfied. Consequently, he recommends that the tests “should be viewed as simply practical rules for action rather than tests having specific probabilities associated with them.” Nelson cautions that “it is possible, though unlikely, for a process to be out of control yet not show any signals from these eight tests.”

Modifying Standard Tests for Special Causes

Some textbooks and references present slightly different versions of Tests 2 and 3. You can use the following options to request these modifications:

- `TEST2RUN=run-length` specifies the length of the pattern for Test 2. The form of the test for each *run-length* is given in the following table. The default *run-length* is 9.

<i>Run-length</i>	Number of Points on One Side of Central Line
7	7 in a row
8	8 in a row
9	9 in a row
11	at least 10 out of 11 in a row
14	at least 12 out of 14 in a row
20	at least 16 out of 20 in a row

- `TEST3RUN=run-length` specifies the length of the pattern for Test 3. The *run-length* values allowed are 6, 7, and 8. The default *run-length* is 6.

The Western Electric Company (now AT&T) *Statistical Quality Control Handbook* and Montgomery (1996) discuss a test that is signaled by eight points in a row in Zone C or beyond (on one side of the central line). You can request this test by specifying `TESTS=2` and `TEST2RUN=8`. The *Handbook* also discusses tests corresponding to Tests 1, 5, 6, 7, and 8.

Kume (1985) recommends a number of tests for special causes that can be regarded as modifications of Tests 2 and 3:

- seven points in a row on one side of the central line. Specify `TESTS=2` and `TEST2RUN=7`.
- at least 10 out of 11 points in a row on one side of the central line. Specify `TESTS=2` and `TEST2RUN=11`.
- at least 12 out of 14 points in a row on one side of the central line. Specify `TESTS=2` and `TEST2RUN=14`.
- at least 16 out of 20 points in a row on one side of the central line. Specify `TESTS=2` and `TEST2RUN=20`.
- seven points in a row steadily increasing or decreasing. Specify `TESTS=3` and `TEST3RUN=7`.

Applying Tests with Varying Subgroup Sample Sizes

See SHWTSC1
in the SAS/QC
Sample Library

Nelson (1989, 1994) describes the use of standardization to apply the tests for special causes to data involving varying subgroup samples. This approach applies the tests to the standardized subgroup statistics, setting the control limits at ± 3 and the zone boundaries at ± 1 and ± 2 . For instance, for an \bar{X} chart with subgroup means \bar{X}_i and varying subgroup sample sizes n_i , the tests are applied to the standardized values $z_i = (\bar{X}_i - \bar{\bar{X}})/(s/\sqrt{n_i})$, where $\bar{\bar{X}}$ estimates the process mean, and s estimates the process standard deviation. You can request this method with the TESTNMETHOD= option,* as illustrated by the following statements:

```

title 'Analysis of Assembly Data';
proc shewhart history=assembly;
  xrchart offset * sample / mu0          = 20
                                sigma0    = 2.24
                                tests      = 1 to 4
                                testnmeth = standardize
                                testlabel  = space
                                ltests     = 2
                                vaxis      = 16 to 26 by 2
                                split      = '/' ;
  label offsetx = 'Avg Offset (cm)/Range';
run;

```

Here the tests are applied to $z_i = (\bar{X}_i - 20)/(2.24/\sqrt{n_i})$. The chart, shown in Figure 48.4, displays the results of the tests on a plot of the *unstandardized* means.




Figure 48.4. The TESTNMETHOD= Option for Varying Subgroup Sizes

*If the TESTNMETHOD= option were omitted in this example, the tests would not be applied, and a warning message would be displayed in the SAS log.

The following statements create an equivalent chart that plots the *standardized* means:

```

data assembly;
  set assembly;
  zx = ( offsetx - 20 ) / ( 2.24 / sqrt( offsetn ) );
run;

title 'Analysis of Standardized Assembly Data';
proc shewhart
  history=assembly (rename = (offsetr=zr offsetn=zn));
  xrchart z * sample / mu0          = 0
                    sigma0         = 2.2361 /* sqrt 5 */
                    limitn          = 5
                    alln
                    tests           = 1 to 4
                    testlabel       = space
                    ltests          = 2
                    vaxis           = -4 to 6 by 2
                    split            = '/' ;
  label zx = 'Std Avg Offset/Range';
run;

```

Here, the SIGMA0= value is the square root of the LIMITN= value. The chart is shown in Figure 48.5.




Figure 48.5. Tests with Standardized Means

Note: In situations where the standard deviation is estimated from the data and the subgroup sample sizes vary, you can use the SMETHOD= option to request various estimation methods, including the method of Burr (1969).

Labeling Signaled Points with a Variable

See SHWTSC1
in the SAS/QC
Sample Library

If a test is signaled at a particular point, the point is labeled by default with the *index* of the test, as illustrated in Figure 48.3.* You can use the TESTLABEL= option to specify a variable in the input data set whose *values* provide the labels, as illustrated by the following statements:

```

title 'Analysis of Assembly Data';
proc shewhart history=assembly;
  xrchart offset * sample / mu0          = 20
                                sigma0    = 2.24
                                limitn     = 5
                                alln
                                tests      = 1 to 4
                                testlabel  = ( comment )
                                ltests     = 2
                                vaxis      = 16 to 24 by 2
                                split      = '/' ;
  label offsetx = 'Avg Offset (cm)/Range';
run;
    
```

The labels are shown in Figure 48.6. It is often helpful to specify a variable with the TESTLABEL= option that provides operator comments or other information that can aid in the identification of special causes.




Figure 48.6. Labeling Points with a TESTLABEL= Variable

*If two or more tests are positive at a particular point, the default label identifies the *index* of the test that was specified first with the TESTS= option.

Applying Tests with Multiple Phases

The data set ASSEMBLY includes a variable named SYSTEM, which indicates the manufacturing system used to produce each assembly. As shown by the following statements, this variable can be temporarily renamed and read as the variable `_PHASE_` to create a control chart that displays the *phases* (groups of consecutive subgroups) for which SYSTEM is equal to T, R, and Q:

See SHWTSC1
in the SAS/QC
Sample Library

```

title 'Manufacturing Systems Used in Assembly';
proc shewhart history=assembly (rename=(system=_phase_));
  xrchart offset * sample /
    mu0          = 20
    sigma0       = 2.24
    limitn       = 5
    alln
    tests        = 1 to 4
    testlabel    = ( comment )
    ltests       = 2
    readphases   = ('T' 'R' 'Q')
    phaselegend
    phaseref
    vaxis        = 16 to 26 by 2
    split        = '/' ;
  label offsetx = 'Avg Offset (cm)/Range';
run;

```

The chart is shown in Figure 48.7.




Figure 48.7. Single Set of Limits with Multiple Phases

Note that a single set of fixed 3σ limits is displayed for all three phases because LIMITN=5 and ALLN are specified. Consequently, the tests requested with the TESTS= option are applied independently of the phases. In general, however, it is possible to display distinct sets of control limits for different phases, and in such situations, the tests are not applied independently of phases, as discussed in the next example.

Applying Tests with Multiple Sets of Control Limits

See SHWTSC2
 in the SAS/QC
 Sample Library

This example is a continuation of the previous example, except that distinct control limits are displayed for each of the phases determined by the variable SYSTEM. The control limit parameters (mean, standard deviation, and nominal sample size) for each phase (manufacturing system) are provided in the following data set:

```
data syslim;
  length _var_ $8 _subgrp_ $8 _type_ $8 _index_ $1;
  input _var_ _subgrp_ _index_ _type_ _mean_ _stddev_
        _limitn_ _sigmas_;
  datalines;
  offset sample R standard 20.5 2.02 5 3
  offset sample Q standard 20.2 2.35 7 3
  offset sample T standard 20.0 2.24 5 3
  ;
```

The following statements read the control limit parameters from SYSLIM and use the READPHASES= and READINDEXES= options to display a distinct set of control limits for each phase:

```
title 'System-Specific Control Limits';
proc shewhart
  limits=syslim
  history=assembly (rename=(system=_phase_));
  xrchart offset * sample /
    tests          = 1 to 4
    testlabel      = ( comment )
    readindexes    = ('T' 'R' 'Q')
    readphases     = ('T' 'R' 'Q')
    phaselegend
    phaseref
    phasebreak
    vaxis          = 16 to 26 by 2
    cinfill        = ligr
    split          = '/' ;
  label offsetx = 'Avg Offset (cm)/Range';
run;
```

The chart is shown in Figure 48.8. The tests requested with the TESTS= option are applied strictly within the phases, since the control limits are not constant across the phases (as in Figure 48.7). In particular, note that the pattern labeled *Reset Tool* in Figure 48.7 is not detected in Figure 48.8.




Figure 48.8. Multiple Sets of Control Limits

In most applications involving multiple control limits, a known change or improvement has occurred at the beginning of each phase; consequently, it is appropriate to restart the tests at the beginning of each phase rather than search for patterns that span the boundaries of consecutive phases. In these situations, the PHASEBREAK option is useful for suppressing the connection of points from one phase to the next. Note that it is not necessary to specify the TESTNMETHOD= option here because the subgroup sample sizes are constant within each phase.

There may be applications in which it is appropriate to apply the tests across phase boundaries. You can use the TESTACROSS option to request this behavior.

```

title 'System-Specific Control Limits';
proc shewhart
  limits=syslim
  history=assembly (rename=(system=_phase_));
  xrchart offset * sample /
    tests          = 1 to 4
    testlabel      = ( comment )
    testnmethode   = standardize
    testacross
    ltests         = 2
    readindexes    = ('T' 'R' 'Q')
    readphases     = ('T' 'R' 'Q')
    phaselegend
    phaseref
    vaxis          = 16 to 26 by 2
    cinfill        = ligr
    split          = '/' ;
  label offsetx = 'Avg Offset (cm)/Range';
run;

```

The chart created with the TESTACROSS option is displayed in Figure 48.9.




Figure 48.9. Multiple Sets of Control Limits with the TESTACROSS Option

Here, it is necessary to specify TESTNMETHOD=STANDARDIZE in conjunction with the TESTACROSS option, since the subgroup sample sizes are not constant across phases.

Although Test 3 is now signaled at sample 18, this result should be interpreted with care since the test is applied to standardized average offsets, and the averages for samples 13, 14, and 15 are standardized differently than the averages for samples 16, 17, and 18. If, for instance, the value of _MEAN_ for phase R in SYSLIM were 21.0 rather than 20.5, the standardized mean for sample 16 would be less than the standardized mean for sample 15, and Test 3 would not be signaled at sample 18.

In summary, when working with multiple control limits, you should

- use the TESTACROSS option only if the process is operating in a continuous manner across phases
- use TESTNMETHOD=STANDARDIZE only if it is clearly understood by users that tests signaled on the chart are based on *standardized* statistics rather than the plotted statistics

Enhancing the Display of Signaled Tests

There are various options for labeling points at which a test is signaled.

- The default label for Test i is *Test i* . See Figure 48.3 on page 1735 for an example.
- Specify TESTLABEL=SPACE to request labels of the form *Test i* . See Figure 48.4 on page 1738 for an example.
- Specify TESTLABEL i =*'label'* to provide a specific *label* for the i^{th} test. See Figure 48.13 on page 1752 for an example.
- Specify TESTLABEL=(*variable*) to request labels provided by a *variable* in the input data set. See Figure 48.6 on page 1740 for an example.

If two or more tests are signaled at a particular point, the label displayed corresponds to the test that was specified first in the TESTS= list.

If you are using a graphics device, you can specify the color of the label and the connecting line segments for the pattern with the CTESTS= option. You can specify the line type for the line segments with the LTESTS= option; see Figure 48.3 on page 1735. If you are using a line printer, you can specify the plot character for the line segments with the TESTCHAR= option.

You can specify the ZONES option to display the zone lines on the chart, and you can specify ZONELABELS to label the zone lines. If you are using a graphics device, you can specify the color of the lines with the CZONES= option, and if you are using a line printer, you can specify the plot character for the lines with the ZONECHAR= option.

Nonstandard Tests for Special Causes

This section describes options and programming techniques for requesting various nonstandard tests for special causes.

Applying Tests to Range and Standard Deviation Charts

See SHWTSC3
in the SAS/QC
Sample Library

If you are using the MRCHART, RCHART, SCHART, XRCHART, or XSCHART statement, you can use the TESTS2= option to request tests for special causes with an R chart or s chart. The syntax and test definitions for the TESTS2= option are identical to those for the TESTS= option, and you can use the ZONES2 and ZONE2LABELS options to display the zones on the secondary chart.

The following statements request Test 1 for a range chart of the data in ASSEMBLY (see page 1734):

```

title 'Analysis of Offset Ranges';
proc shewhart history=assembly;
  rchart offset * sample / sigma0      = 2.24
                                limitn  = 5
                                alln
                                tests2   = 1
                                testlabel = (comment);
  label offsetr = 'Offset Range (cm)';
run;

```

The R chart is shown in Figure 48.10.




Figure 48.10. Range Chart with Test 1

CAUTION: Except for requesting Test 1, use of the TESTS2= option is not recommended for general process control work. At the time of this writing, there is insufficient published research supporting the application of the other tests to R charts and s charts. There are no established guidelines for interpreting the other tests, nor are there assessments of their false signal probabilities or average run length characteristics. The TESTS2= option is intended primarily as a research tool.

Applying Tests Based on Generalized Patterns

In addition to *indices* for standard tests, you can specify up to eight *T-patterns* or *M-patterns* with the TESTS= option:

- Specifying a T-pattern requests a search for k out of m points in a row in the interval (a, b) . Tests based on T-patterns are generalizations of Tests 1, 2, 5, and 6. The average run length properties of tests based on T-patterns have been analyzed by Champ and Woodall (1987). Also refer to Chapter 8 of Wetherill and Brown (1991).
- Specifying an M-pattern requests a search for k points in a row increasing or decreasing. Tests based on M-patterns are generalizations of Test 3.

The general syntax for a T-pattern is of the form

```
T( K=k    M=m    LOWER=a    UPPER=b    SCHEME=scheme
  CODE=character    LABEL='label'    LEGEND='legend' )
```

The options for a T-pattern are summarized in the following table:

Table 48.3. Options for T-Patterns

Option	Description
K=k	number of points ($k \leq m$)
M=m	number of consecutive points
LOWER=value	lower limit of interval (a, b)
UPPER=value	upper limit of interval (a, b)
SCHEME=ONESIDED	one-sided scheme using (a, b)
SCHEME=TWOSIDED	two-sided scheme using $(a, b) \cup (-b, -a)$
CODE=character	identifier for test (A-H)
LABEL='label'	label for points that are signaled
LEGEND='legend'	legend used with the TABLELEGEND option

The following rules apply to the T-pattern options:

1. You must specify SCHEME=*scheme*. Specifying SCHEME=ONESIDED requests a one-sided test that searches for k out of m points in a row in the interval (a, b) . Specifying SCHEME=TWOSIDED with positive values for a and b (where $a < b$) requests a two-sided test that searches for k out of m points in a row in the interval (a, b) or k out of m points in a row in the interval $(-b, -a)$.

2. The values a and b must be specified in standardized units, and they must both have the same sign. For instance, specifying LOWER=2 and UPPER=3 with SCHEME=TWOSIDED corresponds to Zone A in Figure 48.1 on page 1732.
3. Specifying a missing value for the LOWER= option and a negative value for b requests a search in the interval $(-\infty, b)$. Specifying a positive value for a and a missing value for the UPPER= option requests a search in the interval (a, ∞) .
4. You must specify a CODE= *character*, which can be any of the letters A through H. The character identifies the pattern in tables requested with the TABLETESTS and TABLEALL options and in the value of the variable _TESTS_ in the OUTTABLE= data set. The character is analogous to the indices 1 through 8 that are used to identify the standard tests. If you request multiple T-patterns, you must specify a unique character for each pattern.
5. You can specify a *label* with the LABEL= option. The label must be enclosed in quotes and can be up to 16 characters long. The label is used to label points on the chart at which the test defined by the T-pattern is signaled. The LABEL= option is similar to the TESTLABEL n = options used with the standard tests.
6. You must specify a *legend* with the LEGEND= option if you also specify the TABLELEGEND or TABLEALL option. The legend must be enclosed in quotes and can be up to 40 characters long. The legend is used to describe the test defined by the T-pattern in the table legend requested with the TABLELEGEND and TABLEALL options.

See SHWTSC4
in the SAS/QC
Sample Library

An example of a nonstandard test using a T-pattern is the run test based on 14 out of 17 points in a row on the same side of the central line that is suggested by Wheeler and Chambers (1986). The following statements apply this test with Tests 1, 3, and 4. The resulting chart is shown in Figure 48.11.

```

title 'Analysis of Assembly Data';
proc shewhart history=assembly;
  xrchart offset * sample /
    mu0      = 20
    sigma0   = 2.24
    limitn   = 5
    alln
    tests    = 1
             t( k=14 m=17
                lower=0 upper=. scheme=twosided
                code=A label='Test A' )
             3 4
    vaxis    = 16 to 26 by 2
    split    = '//';
  label offsetx = 'Avg Offset (cm)/Range';
run;

```




Figure 48.11. Generalized T-pattern Applied to Assembly Data

The specified T-pattern is signaled at 30th subgroup. Consequently, this point is labeled *Test A*.

The general syntax for an M-pattern is of the form

M(**K**=*k* **DIR**=*direction* **CODE**=*character* **LABEL**=*'label'* **LEGEND**=*'legend'*)

The options for an M-pattern are summarized in the following table:

Table 48.4. Options for M-Patterns

Option	Description
K = <i>k</i>	number of points
DIR =INC	increasing pattern
DIR =DEC	decreasing pattern
CODE = <i>character</i>	identifier for test (A-H)
LABEL = <i>'label'</i>	label for points that are signaled
LEGEND = <i>'legend'</i>	legend used with the TABLELEGEND option

You must specify the direction of the pattern with the DIR= option. The CODE=, LABEL=, and LEGEND= options are used as described on page 1748.

CAUTION: You should not substitute tests based on arbitrarily defined T-patterns and M-patterns for standard tests in general process control applications. The pattern options are intended primarily as a research tool.

See SHWARL2
in the SAS/QC
Sample Library

Champ and Woodall (1990) provide a FORTRAN program for assessing the run length distribution of tests based on T-patterns. A version of their algorithm is implemented by a SAS/IML program in the SAS/QC Sample Library.

If you specify either a T-pattern or M-pattern with the TESTS= option and save the results in an OUTTABLE= data set, the length of the variable _TESTS_ is 16 rather than 8 (the default). The ninth character of _TESTS_ is assigned the value A if the test with CODE=A is signaled, the tenth character of _TESTS_ is assigned the value B if the test with CODE=B is signaled, and so on. If you also specify one or more standard tests, the i^{th} character of _TESTS_ is assigned the value i if Test i is signaled.

Customizing Tests with DATA Step Programs

See SHWTSC5
in the SAS/QC
Sample Library

Occasionally, you may find it necessary to apply customized tests that cannot be specified with the TESTS= option. You can program your own tests as follows:

1. Run the SHEWHART procedure without the TESTS= option and save the results in an OUTTABLE= data set. Use the NOCHART option to suppress the display of the chart.
2. Use a DATA step program to apply your tests to the subgroup statistics in the OUTTABLE= data set. If tests are signaled at certain subgroups, save these results as values of a flag variable named _TESTS_, which should be a character variable of length 8. Recall that each observation of an OUTTABLE= data set corresponds to a subgroup. Assign the character i to the i^{th} character of _TESTS_ if the i^{th} customized test is signaled at that subgroup (otherwise, assign a blank character).
3. Run the procedure reading the modified data set as a TABLE= data set.

The following example illustrates these steps by creating an \bar{X} chart for the data in ASSEMBLY (see page 1734) that signals a special cause of variation if an average is greater than 2.5 standard errors above the central line. The first step is to compute 2.5σ limits and save both the subgroup statistics and the limits in an OUTTABLE= data set named FIRST.

```
proc shewhart history=assembly;
  xchart offset * sample /
    sigmas = 2.5
    outtable = first
    nochart ;

proc print data=first noobs;
run;
```

A partial listing of the data set FIRST is shown in Figure 48.12.

VAR	sample	_SIGMAS_	_LIMITN_	_SUBN_	_LCLX_	_SUBX_	_MEAN_	_UCLX_	_EXLIM_
offset	1	2.5	5	5	18.1515	19.80	20.4733	22.7951	
offset	2	2.5	5	5	18.1515	17.16	20.4733	22.7951	LOWER
offset	3	2.5	5	5	18.1515	20.11	20.4733	22.7951	
offset	4	2.5	5	5	18.1515	20.89	20.4733	22.7951	
offset	5	2.5	5	5	18.1515	20.83	20.4733	22.7951	
offset	6	2.5	5	5	18.1515	18.87	20.4733	22.7951	
offset	7	2.5	5	5	18.1515	20.84	20.4733	22.7951	
offset	8	2.5	5	5	18.1515	23.33	20.4733	22.7951	UPPER
offset	9	2.5	5	5	18.1515	19.21	20.4733	22.7951	
offset	10	2.5	5	5	18.1515	20.48	20.4733	22.7951	
.	
.	
.	
offset	30	2.5	7	7	18.5111	21.32	20.4733	22.4356	

Figure 48.12. Partial Listing of the Data Set FIRST

The second step is to carry out the test and create the flag variable `_TESTS_`.

```
data first;
  set first;
  length _tests_ $ 8;
  if _subx_ > _uclx_ then substr( _tests_, 1 ) = '1';
run;
```

Finally, the data set FIRST is read by the SHEWHART procedure as a TABLE= data set.

```
title 'Customized Analysis of Assembly Data';
proc shewhart table=first;
  xchart offset * sample / tests = 1
          testlabel1 = 'Test Signaled';
  label _subx_ = 'Average Offset (cm)';
run;
```

The chart is shown in Figure 48.13. Note that the variable `_TESTS_` is read “as is” to flag points on the chart, and the standard tests are *not* applied to the data. The option TESTS=1 specifies that a point is to be labeled if the first character of `_TESTS_` for the corresponding subgroup is 1. The label is specified by the TESTLABEL1= option (the default would be *Test1*).




Figure 48.13. Customized Test

In general, you can simultaneously apply up to eight customized tests with the variable `_TESTS_`, which is of length 8. If two or more tests are signaled at a particular point, the label that is displayed corresponds to the test that appears first in the `TESTS=` list. In the preceding example, the test involves only the current subgroup. For customized tests involving patterns that span multiple subgroups, you will find it helpful to use the LAG functions described in *SAS Language Reference: Dictionary*.

Notes:

1. If you provide the variable `_TESTS_` in a `TABLE=` data set, you must also use the `TESTS=` option to specify which characters of `_TESTS_` are to be checked.
2. The `CTESTS=` and `LTESTS=` options specify colors and line styles for *standard* patterns and may not be applicable with customized tests.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC[®] User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

SAS/QC[®] User's Guide, Version 8

Copyright © 1999 SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-493-4

All rights reserved. Printed in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, by any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS[®] and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute in the USA and other countries. [®] indicates USA registration.

IBM[®], ACF/VTAM[®], AIX[®], APPN[®], MVS/ESA[®], OS/2[®], OS/390[®], VM/ESA[®], and VTAM[®] are registered trademarks or trademarks of International Business Machines Corporation. [®] indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.